
    GenObject

"I thought of objects being like biological cells and/or individual computers on a network, only able to communicate with messages (so messaging came at the very beginning -- it took a while to see how to do messaging in a programming language efficiently enough to be useful)." 
— Alan Kay

A library for creating stateful objects backed by GenServer processes with inheritance support.
Elixir's process isolation perfectly embodies Alan Kay's original vision of object-oriented programming. Each process is truly isolated with its own memory space, can only communicate through messages, and fails independently without affecting other processes. GenObject builds on this foundation to provide a clean, object-like interface over GenServer processes, making it easy to create stateful objects that communicate through message passing while maintaining the fault-tolerance and concurrency benefits of the Actor model.
GenObject provides a macro-based DSL for defining object-like structures that maintain state in GenServer processes. Objects support field access, updates, lazy operations, and merging. The library uses the Inherit library to provide powerful inheritance modeling capabilities.
Performance Overhead Warning
GenObjects have significant performance overhead compared to using Elixir's native data structures directly. Each GenObject is backed by a GenServer process, which means every field access, update, or merge operation involves message passing and serialization. This is much slower than working with structs, maps, and other Elixir literals in memory.
GenObjects are best suited for specific patterns where you need long-lived stateful objects, process isolation, fault tolerance, concurrent access to shared state, or complex inheritance hierarchies.
Do not use GenObjects as a general replacement for Elixir's built-in data structures in your application. Use them judiciously where the benefits of process isolation and message passing outweigh the performance costs.
Features
	Stateful Objects: Objects backed by GenServer processes with automatic lifecycle management
	Field Operations: get, set, and merge operations with both synchronous and asynchronous variants
	Lazy Operations: Functions that compute values based on current object state
	Inheritance Support: Inherit library for object inheritance patterns
	Process Safety: All operations are process-safe through GenServer messaging
	Performance: Asynchronous variants for high-performance scenarios

Installation
Add gen_object to your list of dependencies in mix.exs:
def deps do
  [
    {:gen_object, "~> 0.4.0"}
  ]
end
Quick Start
Basic Object Definition
defmodule Person do
  use GenObject, [
    name: "",
    age: nil,
    email: nil
  ]
end

# Create a new person object
person = Person.new(name: "Alice", age: 30)
# Returns: %Person{name: "Alice", age: 30, email: nil, pid: #PID<...>}
Field Access
# Get a single field (most efficient)
name = Person.get(person, :name)
# Returns: "Alice"

# Get multiple fields at once
[name, age] = Person.get(person, [:name, :age])
# Returns: ["Alice", 30]

# Can also use PID directly
age = Person.get(person.pid, :age)
# Returns: 30

# Get the complete object (when you need all fields)
current_person = Person.get(person)
# Returns: %Person{name: "Alice", age: 30, email: nil, pid: #PID<...>}
Field Updates
# Synchronous update (returns updated object)
person = Person.set(person, :age, 31)
person = Person.set(person.pid, :email, "alice@example.com")

# Asynchronous update (returns :ok immediately)
:ok = Person.set!(person, :age, 32)
:ok = Person.set!(person.pid, :name, "Alice Smith")

# Verify async updates
age = Person.get(person, :age)
name = Person.get(person, :name)
Lazy Updates
Lazy updates allow you to compute new values based on the current object state:
# Increment age based on current value
person = Person.set_lazy(person, :age, fn p -> p.age + 1 end)

# Create display name from existing fields
person = Person.set_lazy(person, :display_name, fn p ->
  "#{p.name} (#{p.age})"
end)

# Asynchronous lazy update
:ok = Person.set_lazy!(person, :age, fn p -> p.age + 1 end)
Merging Multiple Fields
# Synchronous merge
person = Person.merge(person, %{
  name: "Alice Johnson",
  age: 35,
  email: "alice.johnson@example.com",
  location: "San Francisco"
})

# Asynchronous merge
:ok = Person.merge!(person, %{age: 36, location: "New York"})

# Lazy merge based on current state
person = Person.merge_lazy(person, fn p ->
  age_group = if p.age < 18, do: "minor", else: "adult"
  %{
    age_group: age_group,
    can_vote: p.age >= 18,
    display_name: "#{p.name} (#{age_group})"
  }
end)
Inheritance with the Inherit Library
GenObject uses the Inherit library to provide powerful inheritance modeling:
Basic Inheritance
defmodule Animal do
  use GenObject, [
    name: "",
    species: "",
    age: 0
  ]
  
  def speak(%__MODULE__{} = animal) do
    "#{animal.name} makes a sound"
  end
end

defmodule Dog do
  use Animal, [
    breed: "",
    trained: false
  ]
  
  # Override parent method
  def speak(%__MODULE__{} = dog) do
    "#{dog.name} barks! Woof!"
  end
  
  def sit(%__MODULE__{trained: true} = dog) do
    Dog.set(dog, :position, :sitting)
  end
  
  def sit(%__MODULE__{trained: false} = dog) do
    {:error, "#{dog.name} is not trained to sit"}
  end
end

# Dog inherits all fields from Animal plus its own
dog = Dog.new(
  name: "Rex", 
  species: "Canis lupus", 
  breed: "Labrador", 
  age: 3,
  trained: true
)

# Use inherited and own methods
Dog.speak(dog)  # "Rex barks! Woof!"
Dog.sit(dog)    # Updates position to :sitting
Multi-level Inheritance
defmodule LivingThing do
  use GenObject, [
    alive: true,
    birth_date: nil
  ]
end

defmodule Animal do
  use LivingThing, [
    name: "",
    species: ""
  ]
end

defmodule Mammal do
  use Animal, [
    warm_blooded: true,
    fur_color: nil
  ]
end

defmodule Dog do
  use Mammal, [
    breed: "",
    trained: false
  ]
end

# Dog inherits from the entire chain
dog = Dog.new(
  name: "Buddy",
  species: "Canis lupus",
  breed: "Golden Retriever",
  fur_color: "golden",
  birth_date: ~D[2020-01-15],
  trained: true
)
Complex Inheritance Patterns
defmodule Vehicle do
  use GenObject, [
    make: "",
    model: "",
    year: nil,
    mileage: 0
  ]
  
  def drive(%__MODULE__{} = vehicle, distance) do
    Vehicle.set_lazy(vehicle, :mileage, fn v -> v.mileage + distance end)
  end
end

defmodule Car do
  use Vehicle, [
    doors: 4,
    fuel_type: :gasoline
  ]
  
  def honk(%__MODULE__{} = car) do
    "#{car.make} #{car.model} honks: BEEP BEEP!"
  end
end

defmodule ElectricCar do
  use Car, [
    battery_capacity: 0,
    charge_level: 100,
    fuel_type: :electric  # Override parent default
  ]
  
  def charge(%__MODULE__{} = car, amount) do
    ElectricCar.set_lazy(car, :charge_level, fn c ->
      min(100, c.charge_level + amount)
    end)
  end
  
  # Override parent method
  def drive(%__MODULE__{} = car, distance) do
    car = super(car, distance)  # Call parent implementation
    # Reduce charge based on distance
    ElectricCar.set_lazy(car, :charge_level, fn c ->
      max(0, c.charge_level - div(distance, 10))
    end)
  end
end

# Create an electric car with full inheritance chain
tesla = ElectricCar.new(
  make: "Tesla",
  model: "Model 3",
  year: 2023,
  battery_capacity: 75,
  doors: 4
)

# Use methods from all levels of inheritance
tesla = ElectricCar.drive(tesla, 100)    # Inherited and overridden
tesla = ElectricCar.charge(tesla, 20)    # Own method
message = Car.honk(tesla)                # Parent method
Virtual Attributes
GenObject supports virtual attributes - computed fields that don't store data directly but calculate values dynamically or transform input into multiple real fields. Virtual attributes are implemented by overriding the handle_get/2, handle_set/2, and handle_merge/2 callbacks.
Computed Virtual Attributes
Virtual attributes that calculate values from existing fields:
defmodule Person do
  use GenObject, [
    first_name: "",
    last_name: "",
    age: nil
  ]
  
  # Virtual attribute that computes full name from parts
  def handle_get(:name, %Person{} = person) do
    "#{person.first_name} #{person.last_name}"
  end
  
  # Fall back to default behavior for regular fields
  def handle_get(field, object) do
    super(field, object)
  end
end

person = Person.new(first_name: "Alice", last_name: "Smith")
Person.get(person, :name)  # Returns: "Alice Smith"

# Get multiple fields including virtual attributes
[name, first_name, last_name] = Person.get(person, [:name, :first_name, :last_name])
# Returns: ["Alice Smith", "Alice", "Smith"]
Input-Transforming Virtual Attributes
Virtual attributes that parse input and set multiple real fields:
defmodule Person do
  use GenObject, [
    first_name: "",
    last_name: "",
    age: nil
  ]
  
  # Virtual attribute that splits full name into parts
  def handle_set({:name, full_name}, %Person{} = person) do
    [first_name, last_name] = String.split(full_name, " ", parts: 2)
    Map.merge(person, %{first_name: first_name, last_name: last_name})
  end

  # Fall back to default behavior for other operations
  def handle_set(pair, object), do: super(pair, object)
  
  # Computed virtual attribute (from above)
  def handle_get(:name, %Person{} = person) do
    "#{person.first_name} #{person.last_name}"
  end
  
  # Fall back to default behavior for other operations
  def handle_get(field, object), do: super(field, object)
end

person = Person.new()
person = Person.set(person, :name, "Bob Johnson")

[first_name, last_name, name] = Person.get(person, [:first_name, :last_name, :name])
# Returns: ["Bob", "Johnson", "Bob Johnson"]

# Virtual attributes work with merge operations too
person = Person.merge(person, %{name: "Carol Davis", age: 25})
[first_name, last_name] = Person.get(person, [:first_name, :last_name])
# Returns: ["Carol", "Davis"]
Advanced Virtual Attribute Patterns
You can create more complex virtual attributes for validation, formatting, or derived data:
defmodule BankAccount do
  use GenObject, [
    balance_cents: 0,
    account_number: "",
    routing_number: ""
  ]
  
  # Virtual attribute for balance in dollars
  def handle_get(:balance, %BankAccount{} = account) do
    account.balance_cents / 100
  end
  
  # Virtual attribute for formatted account info
  def handle_get(:account_info, %BankAccount{} = account) do
    "Account: #{account.account_number} (Routing: #{account.routing_number})"
  end

  def handle_get(field, object), do: super(field, object)
  
  def handle_set({:balance, dollars}, %BankAccount{} = account) do
    cents = round(dollars * 100)
    Map.put(account, :balance_cents, cents)
  end
  
  # Validation virtual attribute
  def handle_set({:account_details, details}, %BankAccount{} = account) do
    %{account_number: acct, routing_number: routing} = details
    
    unless valid_account_number?(acct) do
      raise ArgumentError, "Invalid account number"
    end
    
    Map.merge(account, %{account_number: acct, routing_number: routing})
  end
  
  def handle_set(pair, object), do: super(pair, object)
  
  defp valid_account_number?(number) do
    String.length(number) >= 8
  end
end

account = BankAccount.new(balance_cents: 10050)
balance = BankAccount.get(account, :balance)  # Returns: 100.5

account = BankAccount.set(account, :balance, 250.75)
[balance, balance_cents] = BankAccount.get(account, [:balance, :balance_cents])
# Returns: [250.75, 25075]
Advanced Usage
Custom GenServer Callbacks
You can override GenServer callbacks while still using GenObject functionality:
defmodule TimestampedObject do
  use GenObject, [
    data: nil,
    created_at: nil,
    updated_at: nil
  ]
  
  # Override init to set timestamps
  def init(opts) do
    now = DateTime.utc_now()
    opts = opts
    |> Keyword.set(:created_at, now)
    |> Keyword.set(:updated_at, now)
    
    super(opts)  # Call GenObject's init
  end
  
  # Override handle_call to update timestamps
  def handle_call({:set, field, value}, from, object) do
    result = super({:set, field, value}, from, object)
    case result do
      {:reply, updated_object, state} ->
        updated_state = TimestampedObject.set(state, :updated_at, DateTime.utc_now())
        {:reply, updated_object, updated_state}
      other -> other
    end
  end
end
Supervision
GenObjects can be supervised like any GenServer:
defmodule MyApp.ObjectSupervisor do
  use Supervisor
  
  def start_link(_opts) do
    Supervisor.start_link(__MODULE__, [], name: __MODULE__)
  end
  
  def init([]) do
    children = [
      {Person, [name: "Default Person"]},
      {Dog, [name: "Default Dog", breed: "Mixed"]}
    ]
    
    Supervisor.init(children, strategy: :one_for_one)
  end
end
Performance Considerations
	Use asynchronous operations (set!/3, merge!/2, etc.) when you don't need the return value
	Use get/2 for single fields instead of get/1 when you only need one field
	Batch multiple updates using merge/2 instead of multiple set/3 calls
	Lazy operations are computed synchronously, so complex computations may block

API Reference
Creation and Lifecycle
	YourModule.new/1 - Create a new object
	YourModule.close/1 - Stop the object process

Field Access
	YourModule.get/1 - Get complete object state
	YourModule.get/2 - Get specific field value or multiple field values

Field Updates
	YourModule.set/3 - Update field synchronously
	YourModule.set!/3 - Update field asynchronously
	YourModule.set_lazy/3 - Update field with function synchronously
	YourModule.set_lazy!/3 - Update field with function asynchronously

Bulk Operations
	YourModule.merge/2 - Merge multiple fields synchronously
	YourModule.merge!/2 - Merge multiple fields asynchronously
	YourModule.merge_lazy/2 - Merge with function synchronously
	YourModule.merge_lazy!/2 - Merge with function asynchronously

All functions accept either a PID or an object struct containing a :pid field. Replace YourModule with your actual module name (e.g., Person, Dog, etc.).
Contributing
	Fork the repository
	Create your feature branch (git checkout -b my-new-feature)
	Commit your changes (git commit -am 'Add some feature')
	Push to the branch (git push origin my-new-feature)  
	Create a new Pull Request

License
This project is licensed under the MIT License - see the LICENSE.md file for details.


  

    MIT License

Copyright (c) 2024 DockYard, Inc.
Permission is hereby granted, free of charge, to any person obtaining a copy
of this software and associated documentation files (the "Software"), to deal
in the Software without restriction, including without limitation the rights
to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
copies of the Software, and to permit persons to whom the Software is
furnished to do so, subject to the following conditions:
The above copyright notice and this permission notice shall be included in all
copies or substantial portions of the Software.
THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE
SOFTWARE.


  

    
GenObject 
    



      
A library for creating stateful objects backed by GenServer processes with inheritance support.
GenObject provides a macro-based DSL for defining object-like structures that maintain
state in GenServer processes. Objects support field access, updates, lazy operations,
and merging. The library integrates with the Inherit
library to provide inheritance modeling capabilities.
Features
	Stateful Objects: Objects backed by GenServer processes with automatic lifecycle management
	Field Operations: get, set, and merge operations with both synchronous and asynchronous variants
	Lazy Operations: Functions that compute values based on current object state
	Inheritance Support: Integration with the Inherit library for object inheritance patterns
	Process Safety: All operations are process-safe through GenServer messaging

Quick Start
defmodule Person do
  use GenObject, [
    name: "",
    age: nil,
    email: nil
  ]
end

# Create a new person object
person = Person.new(name: "Alice", age: 30)

# Access fields
Person.get(person, :first_name)  # "Alice"

# Update a single field
person = Person.set(person, :age, 31)

# Update multiple fields
person = Person.merge(person, %{name: "Alice Smith", email: "alice@example.com"})

# Lazy updates based on current state
person = Person.set_lazy(person, :age, fn p -> p.age + 1 end)
Inheritance with the Inherit Library
GenObject integrates seamlessly with the Inherit library
to provide object inheritance patterns. The Inherit library allows you to define parent-child
relationships between objects and inherit fields and behaviors.
defmodule Animal do
  use GenObject, [
    name: "",
    species: ""
  ]
end

defmodule Dog do
  use Animal, [
    breed: "",
    trained: false
  ]
end

# Dog inherits all fields from Animal plus its own
dog = Dog.new(name: "Rex", species: "Canis lupus", breed: "Labrador")

        get(pid)

      


        Retrieves the complete current state of an object.



    


    
      
        get(pid, field_or_fields)

      


        Retrieves the value of a specific field or multiple fields from an object.



    


    
      
        handle_get(field, object)

      


        Handles getting a field value from an object.



    


    
      
        handle_merge(fields, object)

      


        Handles merging multiple field values into an object.



    


    
      
        handle_set(pair, object)

      


        Handles setting a field value in an object.



    


    
      
        merge(pid, fields)

      


        Merges multiple fields into an object and returns the updated object struct.



    


    
      
        merge!(pid, fields)

      


        Merges multiple fields into an object asynchronously, returning :ok immediately.



    


    
      
        merge_lazy(pid, func)

      


        Merges multiple fields using a function that computes values based on current object state.



    


    
      
        merge_lazy!(pid, func)

      


        Merges multiple fields using a function asynchronously, returning :ok immediately.



    


    
      
        new(opts \\ [])

      


        Create a new object with the specified fields



    


    
      
        set(pid, field, value)

      


        Updates a specific field in an object and returns the updated object struct.



    


    
      
        set!(pid, field, value)

      


        Updates a specific field in an object asynchronously and returns :ok immediately.



    


    
      
        set_lazy(pid, field, func)

      


        Updates a specific field using a function that computes the new value based on current object state.



    


    
      
        set_lazy!(pid, field, func)

      


        Updates a specific field using a function asynchronously, returning :ok immediately.



    


    
      
        stop(pid)

      


        Gracefully stops the curent object's GenServer

Retrieves the complete current state of an object.
Returns the full object struct containing all fields and their current values.
Accepts either a PID directly or a struct containing a :pid field.
Parameters
	pid_or_object - The PID of the GenObject process, or an object struct containing a :pid field

Examples
# Using the object struct
person = Person.new(name: "Alice", age: 30)
current_state = GenObject.get(person)
# Returns: %Person{name: "Alice", age: 30, pid: #PID<...>}

# Using the PID directly  
current_state = GenObject.get(person.pid)
# Returns: %Person{name: "Alice", age: 30, pid: #PID<...>}

Retrieves the value of a specific field or multiple fields from an object.
When given a single field (atom), returns the current value of that field.
When given a list of fields, returns a list of values in the same order as requested.
This is more efficient than retrieving the entire object struct when you only need specific fields.
Parameters
	pid_or_object - The PID of the GenObject process, or an object struct containing a :pid field
	field_or_fields - Either an atom representing a single field name, or a list of atoms for multiple fields

Examples
person = Person.new(first_name: "Alice", last_name: "Smith", age: 30)

# Get a single field
first_name = Person.get(person, :first_name)
# Returns: "Alice"

# Get multiple fields at once
[first_name, age] = Person.get(person, [:first_name, :age])
# Returns: ["Alice", 30]

# Works with PIDs too
[last_name, age] = Person.get(person.pid, [:last_name, :age])
# Returns: ["Smith", 30]

# Virtual attributes work with lists too
[name, age] = Person.get(person, [:name, :age])
# Returns: ["Alice Smith", 30]

Handles getting a field value from an object.
This function is called whenever a field is accessed via get/2. Override this
function to implement virtual attributes that compute values dynamically instead
of storing them directly in the object state.
Parameters
	field - The atom representing the field name being accessed
	object - The current object struct

Returns
The value for the requested field. For regular fields, this returns the stored
value. For virtual attributes, this can return any computed value.
Examples
# Define a virtual attribute that combines first and last name
def handle_get(:name, %Person{} = person) do
  "#{person.first_name} #{person.last_name}"
end

# Fall back to default behavior for other fields  
def handle_get(field, object) do
  super(field, object)
end

Handles merging multiple field values into an object.
This function is called whenever multiple fields are updated via merge/2, merge!/2,
or their lazy variants. The default implementation processes each field individually
through handle_set/2, which means virtual attributes work automatically with merge
operations.
Override this function if you need custom logic for processing multiple fields together,
such as validating field combinations or applying transformations that depend on
multiple input values.
Parameters
	fields - A map of field-value pairs to merge into the object
	object - The current object struct

Returns
The updated object struct with all field changes applied.
Examples
# Custom merge logic that validates address fields together
def handle_merge(%{street: _, city: _, zip: _} = address_fields, %Person{} = person) do
  # Custom validation logic here
  if valid_address?(address_fields) do
    super(address_fields, person)
  else
    raise "Invalid address combination"
  end
end

# Fall back to default behavior for other merges
def handle_merge(fields, object) do
  super(fields, object)
end

Handles setting a field value in an object.
This function is called whenever a field is updated via put/3, put!/3, or their
lazy variants. Override this function to implement virtual attributes that can parse
or transform input values into multiple real fields.
Parameters
	pair - A tuple of {field, value} representing the field name and new value
	object - The current object struct

Returns
The updated object struct with the field changes applied.
Examples
# Define a virtual attribute that splits a full name into parts
def handle_set({:name, full_name}, %Person{} = person) do
  [first_name, last_name] = String.split(full_name, " ", parts: 2)
  Map.merge(person, %{first_name: first_name, last_name: last_name})
end

# Fall back to default behavior for other fields
def handle_set(pair, object) do
  super(pair, object)
end

Merges multiple fields into an object and returns the updated object struct.
This synchronous operation updates multiple fields simultaneously, similar to struct/2
but for live GenObject processes. More efficient than multiple individual set/3 calls
when updating several fields at once.
Parameters
	pid_or_object - The PID of the GenObject process, or an object struct containing a :pid field
	fields - A map of field-value pairs to merge into the object

Examples
person = Person.new(name: "Alice", age: 30)

# Merge multiple fields using the object struct
updated_person = GenObject.merge(person, %{
  name: "Alice Smith",
  age: 31,
  email: "alice.smith@example.com"
})
# Returns: %Person{name: "Alice Smith", age: 31, email: "alice.smith@example.com", pid: #PID<...>}

# Merge using the PID directly
updated_person = GenObject.merge(person.pid, %{age: 32, location: "New York"})

Merges multiple fields into an object asynchronously, returning :ok immediately.
This is the asynchronous version of merge/2. It sends a cast message to update
multiple fields simultaneously but returns immediately without waiting for the
operation to complete. Use this for better performance when you don't need the updated struct.
Parameters
	pid_or_object - The PID of the GenObject process, or an object struct containing a :pid field
	fields - A map of field-value pairs to merge into the object

Examples
person = Person.new(name: "Alice", age: 30)

# Async merge multiple fields
:ok = GenObject.merge!(person, %{
  name: "Alice Smith",
  age: 31,
  email: "alice.smith@example.com"
})

# Verify the updates were applied
updated_person = GenObject.get(person)
# Returns: %Person{name: "Alice Smith", age: 31, email: "alice.smith@example.com", pid: #PID<...>}

Merges multiple fields using a function that computes values based on current object state.
This synchronous operation allows you to merge multiple fields using a function that
receives the current object state and returns a map of field-value pairs to merge.
Useful for complex updates that depend on multiple fields or computed values.
Parameters
	pid_or_object - The PID of the GenObject process, or an object struct containing a :pid field
	func - A function that takes the current object struct and returns a map of field-value pairs to merge

Examples
person = Person.new(name: "Alice", age: 30)

# Merge fields based on current state
updated_person = GenObject.merge_lazy(person, fn p -> 
  %{
    name: p.name <> " Smith",
    age: p.age + 1,
    display_name: p.name <> " (" <> Integer.to_string(p.age + 1) <> ")"
  }
end)
# Returns: %Person{name: "Alice Smith", age: 31, display_name: "Alice (31)", pid: #PID<...>}

# Complex computation based on multiple fields
updated_person = GenObject.merge_lazy(person.pid, fn p ->
  age_group = if p.age < 18, do: "minor", else: "adult"
  %{age_group: age_group, can_vote: p.age >= 18}
end)

Merges multiple fields using a function asynchronously, returning :ok immediately.
This is the asynchronous version of merge_lazy/2. It sends a cast message to merge
multiple fields using a function that computes values based on current object state,
but returns immediately without waiting for the operation to complete.
Parameters
	pid_or_object - The PID of the GenObject process, or an object struct containing a :pid field
	func - A function that takes the current object struct and returns a map of field-value pairs to merge

Examples
person = Person.new(name: "Alice", age: 30)

# Async merge fields based on current state
:ok = GenObject.merge_lazy!(person, fn p -> 
  %{
    name: p.name <> " Smith",
    age: p.age + 1,
    display_name: p.name <> " (" <> Integer.to_string(p.age + 1) <> ")"
  }
end)

# Verify the updates were applied
updated_person = GenObject.get(person)
# Returns: %Person{name: "Alice Smith", age: 31, display_name: "Alice (31)", pid: #PID<...>}

Create a new object with the specified fields

Updates a specific field in an object and returns the updated object struct.
This is a synchronous operation that updates the field value and returns the complete
updated object struct. The operation is atomic and thread-safe.
Parameters
	pid_or_object - The PID of the GenObject process, or an object struct containing a :pid field
	field - The atom representing the field name to update
	value - The new value to set for the field

Examples
person = Person.new(name: "Alice", age: 30)

# Update using the object struct
updated_person = GenObject.set(person, :age, 31)
# Returns: %Person{name: "Alice", age: 31, pid: #PID<...>}

# Update using the PID directly
updated_person = GenObject.set(person.pid, :first_name, "Alice Smith")
# Returns: %Person{name: "Alice Smith", age: 31, pid: #PID<...>}

Updates a specific field in an object asynchronously and returns :ok immediately.
This is an asynchronous operation that sends a cast message to update the field
and returns immediately without waiting for confirmation. Use this when you don't
need the updated object struct and want better performance.
Parameters
	pid_or_object - The PID of the GenObject process, or an object struct containing a :pid field
	field - The atom representing the field name to update
	value - The new value to set for the field

Examples
person = Person.new(name: "Alice", age: 30)

# Async update using the object struct
:ok = GenObject.set!(person, :age, 31)

# Async update using the PID directly
:ok = GenObject.set!(person.pid, :first_name, "Alice Smith")

# Verify the update was applied
updated_person = GenObject.get(person)
# Returns: %Person{name: "Alice Smith", age: 31, pid: #PID<...>}

Updates a specific field using a function that computes the new value based on current object state.
This synchronous operation allows you to update a field using a function that receives
the current object state and returns the new value for the field. Useful for updates
that depend on the current state of the object.
Parameters
	pid_or_object - The PID of the GenObject process, or an object struct containing a :pid field
	field - The atom representing the field name to update
	func - A function that takes the current object struct and returns the new value for the field

Examples
person = Person.new(name: "Alice", age: 30)

# Increment age based on current value
updated_person = GenObject.set_lazy(person, :age, fn p -> p.age + 1 end)
# Returns: %Person{name: "Alice", age: 31, pid: #PID<...>}

# Modify name based on current state
updated_person = GenObject.set_lazy(person.pid, :first_name, fn p -> 
  p.name <> " (" <> Integer.to_string(p.age) <> ")"
end)
# Returns: %Person{name: "Alice (30)", age: 30, pid: #PID<...>}

Updates a specific field using a function asynchronously, returning :ok immediately.
This is the asynchronous version of set_lazy/3. It sends a cast message to update
the field using a function that computes the new value based on current object state,
but returns immediately without waiting for the operation to complete.
Parameters
	pid_or_object - The PID of the GenObject process, or an object struct containing a :pid field
	field - The atom representing the field name to update
	func - A function that takes the current object struct and returns the new value for the field

Examples
person = Person.new(name: "Alice", age: 30)

# Async increment age based on current value
:ok = GenObject.set_lazy!(person, :age, fn p -> p.age + 1 end)

# Async modify name based on current state
:ok = GenObject.set_lazy!(person.pid, :first_name, fn p -> 
  p.name <> " (" <> Integer.to_string(p.age) <> ")"
end)

# Verify the updates were applied
updated_person = GenObject.get(person)

Gracefully stops the curent object's GenServer

